
	Reason for change:
	In both these of test cases, there are multiple REGISTER / 401 Unauthorised transactions before a protected REGISTER message is received. Both test cases can have issues when UDP transport is used due to the correct port information not being configured at IP PTC.

1) In test case 9.1 on receiving the first register the ports and security information is configured at IP PTC. On receiving the 2nd REGISTER (step 3) this information is not refreshed. If the UE is using UDP transport the port number on which it expectes to receive a response (indicated in the contact header) may have changed from the first REGISTER. This can lead to the 401 Unauthorised response (step 4) being sent to the wrong UDP port and not processed by the UE.
2) Similarly for 9.2, on receiving the first REGISTER message no configuration is sent to IP PTC. This can also mean the response is sent on the wrong port when UDP transport is being used.

	
	

	Summary of change:
	1. For 9.1, send “PortsAndSecurityConfig” to IP PTC on receiving the 2nd REGISTER, containing UDP port details but without any security context. Note  The full security configuration from this REGISTER is not configured here because IP PTC has a limt that it can only be configured with 2 security contexts so an error would be thrown after the 3rd REGISTER recived at step 6.
2. For test case 9.2, send “PortsAndSecurityConfig” to IP PTC on receiving the 2nd REGISTER, containing UDP port details but without any security context.

	
	

	Consequences if not approved:
	A conformant UE may fail these test cases if it uses UDP transport for REGISTER requests

	Other comments:
	


Change 1 
	Function name
	f_TC_9_1_IMS()

	Reason for change
	In test case 9.1 on receiving the first register the ports and security information is configured at IP PTC. On receiving the 2nd REGISTER (step 3) this information is not refreshed. If the UE is using UDP transport the port number on which it expectes to receive a response (indicated in the contact header) may have changed from the first REGISTER. This can lead to the 401 Unauthorised response (step 4) being sent to the wrong UDP port and not processed by the UE.

	Summary of change
	For 9.1, send “PortsAndSecurityConfig” to IP PTC on receiving the 2nd REGISTER, containing UDP port details but without any security context. Note  The full security configuration from this REGISTER is not configured here because IP PTC has a limt that it can only be configured with 2 security contexts so an error would be thrown after the 3rd REGISTER recived at step 6.

	TTCN module
	IMS_CC_AuthenticationTestcases.ttcn

	MCC160 Comment
	


Before change

     function f_TC_9_1_IMS() runs on IMS_PTC

  { /* @sic R5-153463: step 6 (403 forbidden) replaced by step 6..12 (completition of registration procedure) sic@ */

    var charstring v_InvalidNonce := fl_GenerateInvalidNonce(macError);    // Nonce with invalid MAC

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var REGISTER_Request v_RegisterReq;

    var charstring v_CallId;

    var integer i;

    var SecurityClientParamsList_Type v_SecurityClientParamsList;

    var template CommaParam_List v_DigestResponseContainingAuts := {cr_GenericParam("auts", ?), *};

    f_IMS_CC_Preamble(IPCAN_SignallingOnly, IMS_NULL, IMS_Security);

    f_IMS_CC_StartSignalling(IPCAN_InitialRegistration);

    f_IMS_TestBody_Set(true);

    // Step 1

    // @siclog "Step 1" siclog@

    v_IMS_DATA_REQ := f_IMS_REGISTER_InitialRequest();

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    v_CallId := v_RegisterReq.msgHeader.callId.callid;

    f_IMS_Register_SecurityInit(v_RegisterReq);   // @sic R5s120947 sic@

    v_SecurityClientParamsList[0] := f_IMS_Register_GetSecurityClientParams(v_RegisterReq);  /* @sic R5-144754 sic@ */

    for (i:=0; i<2; i:=i+1) {

      // Step 2/4

      // @siclog "Step 2/4" siclog@

      IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterReq, v_InvalidNonce))));

      // Step 3/5

      // @siclog "Step 3/5" siclog@

      v_IMS_DATA_REQ := f_IMS_REGISTER_RequestWithSpecificAuth(v_CallId, cr_Authorization_9_1);

      v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

      v_SecurityClientParamsList := fl_CheckSecurityClientParams(v_SecurityClientParamsList, v_RegisterReq);  /* @sic R5-144754 sic@ */

      // Check that digestResponse does not contain (non-empty) "auts" parameter

      if (match(v_RegisterReq.msgHeader.authorization.body[0].digestResponse, v_DigestResponseContainingAuts)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "");

      } else {

        f_IMS_PreliminaryPass(__FILE__, __LINE__, "");

      }

    }

    // @siclog "Step 6" siclog@

    f_IMS_Register_SecurityInit(v_RegisterReq);
    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterReq))));

    // @siclog "Step 7" siclog@

    v_IMS_DATA_REQ := f_IMS_REGISTER_SubsequentRequest();

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);   /* security has changed to "protected" now */

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    // @siclog "Step 8" siclog@

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_RegisterReq))));

    // @siclog "Step 9 - 12" siclog@

    f_IMS_Register_SubscribeNotify(v_RegisterReq);

    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_InitialRegistration);

  }
After change

function f_TC_9_1_IMS() runs on IMS_PTC

  { /* @sic R5-153463: step 6 (403 forbidden) replaced by step 6..12 (completition of registration procedure) sic@ */

    var charstring v_InvalidNonce := fl_GenerateInvalidNonce(macError);    // Nonce with invalid MAC

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var REGISTER_Request v_RegisterReq;

    var charstring v_CallId;

    var integer i;

    var SecurityClientParamsList_Type v_SecurityClientParamsList;

    var template CommaParam_List v_DigestResponseContainingAuts := {cr_GenericParam("auts", ?), *};

    f_IMS_CC_Preamble(IPCAN_SignallingOnly, IMS_NULL, IMS_Security);

    f_IMS_CC_StartSignalling(IPCAN_InitialRegistration);

    f_IMS_TestBody_Set(true);

    // Step 1

    // @siclog "Step 1" siclog@

    v_IMS_DATA_REQ := f_IMS_REGISTER_InitialRequest();

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    v_CallId := v_RegisterReq.msgHeader.callId.callid;

    f_IMS_Register_SecurityInit(v_RegisterReq);   // @sic R5s120947 sic@

    v_SecurityClientParamsList[0] := f_IMS_Register_GetSecurityClientParams(v_RegisterReq);  /* @sic R5-144754 sic@ */

    for (i:=0; i<2; i:=i+1) {

      // Step 2/4

      // @siclog "Step 2/4" siclog@

      IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterReq, v_InvalidNonce))));

      // Step 3/5

      // @siclog "Step 3/5" siclog@

      v_IMS_DATA_REQ := f_IMS_REGISTER_RequestWithSpecificAuth(v_CallId, cr_Authorization_9_1);

      v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

      v_SecurityClientParamsList := fl_CheckSecurityClientParams(v_SecurityClientParamsList, v_RegisterReq);  /* @sic R5-144754 sic@ */

      // Check that digestResponse does not contain (non-empty) "auts" parameter

      if (match(v_RegisterReq.msgHeader.authorization.body[0].digestResponse, v_DigestResponseContainingAuts)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "");

      } else {

        f_IMS_PreliminaryPass(__FILE__, __LINE__, "");

      }
      //2nd REGISTER - send port configuration ONLY

      //3rd REGISTER - send port and security configurations

      //this is because IP PTC cannot handle 3 security contexts! 

      if (i==0) {

           f_IMS_PortsAndSecurityConfig(f_SIP_ContactUrl_GetPort(f_MessageHeader_GetContactSipUrl(v_RegisterReq.msgHeader)),                 // may be omit in which case 5060 is chosen by the IP PTC

                                   cs_IMS_RegistrationInfo(f_IMS_PTC_NW_Address_Get(), f_IMS_PTC_UE_Address_Get()));

      }

      else

      {

          f_IMS_Register_SecurityInit(v_RegisterReq);

      }
    }

    // @siclog "Step 6" siclog@

    //MOVED f_IMS_Register_SecurityInit(v_RegisterReq);
    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterReq))));

    // @siclog "Step 7" siclog@

    v_IMS_DATA_REQ := f_IMS_REGISTER_SubsequentRequest();

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);   /* security has changed to "protected" now */

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    // @siclog "Step 8" siclog@

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_RegisterReq))));

    // @siclog "Step 9 - 12" siclog@

    f_IMS_Register_SubscribeNotify(v_RegisterReq);

    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_InitialRegistration);

  }
Change 2
	Function name
	f_TC_9_2_IMS()

	Reason for change
	For 9.2, on receiving the first REGISTER message no configuration is sent to IP PTC. This can also mean the response is sent on the wrong port when UDP transport is being used.

	Summary of change
	2.
For test case 9.2, send “PortsAndSecurityConfig” to IP PTC on receiving the 2nd REGISTER, containing UDP port details but without any security context.

	TTCN module
	IMS_CC_AuthenticationTestcases.ttcn

	MCC160 Comment
	


Before change

	  function f_TC_9_2_IMS() runs on IMS_PTC

  {

    var charstring v_InvalidNonce := fl_GenerateInvalidNonce(sqnFailure);

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var Common_AuthenticationParams_Type v_AuthenticationParams;

    var SecurityClientParamsList_Type v_SecurityClientParamsList;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var REGISTER_Request v_RegisterReq;

    var charstring v_CallId;

    f_IMS_CC_Preamble(IPCAN_SignallingOnly, IMS_NULL, IMS_Security);

    f_IMS_CC_StartSignalling(IPCAN_InitialRegistration);

    f_IMS_TestBody_Set(true);

    // Step 1

    // @siclog "Step 1" siclog@

    v_IMS_DATA_REQ := f_IMS_REGISTER_InitialRequest();

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    v_CallId := v_RegisterReq.msgHeader.callId.callid;

    v_SecurityClientParamsList[0] := f_IMS_Register_GetSecurityClientParams(v_RegisterReq);  /* @sic R5-144754 sic@ */

    // Step 2

    // @siclog "Step 2" siclog@

    /* f_IMS_Register_SecurityInit(v_RegisterReq);       @sic R5s130264 Change 2.1: security needs to be initialised just before (first) 401 response is sent sic@  */

    v_AuthenticationParams := f_IMS_AuthenticationInit(tsc_IMS_AuthRAND);

    f_IMS_Register_SecurityInitPortsAndSPIs(v_RegisterReq, v_AuthenticationParams);   /* @sic R5s131004 - MCC160 implementation  sic@

                                                                                         security needs to be initialised at the IMS PTC otherwise f_IMS_PTC_BuildSecurityServerParam will cause a runtime errror */

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterReq, v_InvalidNonce))));

    // Step 3

    // @siclog "Step 3" siclog@

    v_IMS_DATA_REQ := f_IMS_REGISTER_RequestWithSpecificAuth(v_CallId, cr_Authorization_9_2(v_InvalidNonce, tsc_IMS_Opaque));

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    v_SecurityClientParamsList := fl_CheckSecurityClientParams(v_SecurityClientParamsList, v_RegisterReq);  /* @sic R5-144754 sic@ */

    // Step 4

    // @siclog "Step 4" siclog@

    f_IMS_Register_SecurityInit(v_RegisterReq);       /* @sic R5s130573: IPSec layer needs to be updated acc. to new Security client header sent by the UE sic@ */

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterReq))));

    // Step 5

    // @siclog "Step 5" siclog@

    v_IMS_DATA_REQ := f_IMS_REGISTER_SubsequentRequest();

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);   /* @sic R5s130264 Change 2.2: security has changed to "protected" now sic@ */

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    // Step 6

    // @siclog "Step 6" siclog@

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_RegisterReq))));

    f_IMS_Register_SubscribeNotify(v_RegisterReq);

    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_InitialRegistration);

  }

}


After change
	function f_TC_9_2_IMS() runs on IMS_PTC

  {

    var charstring v_InvalidNonce := fl_GenerateInvalidNonce(sqnFailure);

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var Common_AuthenticationParams_Type v_AuthenticationParams;

    var SecurityClientParamsList_Type v_SecurityClientParamsList;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var REGISTER_Request v_RegisterReq;

    var charstring v_CallId;

    f_IMS_CC_Preamble(IPCAN_SignallingOnly, IMS_NULL, IMS_Security);

    f_IMS_CC_StartSignalling(IPCAN_InitialRegistration);

    f_IMS_TestBody_Set(true);

    // Step 1

    // @siclog "Step 1" siclog@

    v_IMS_DATA_REQ := f_IMS_REGISTER_InitialRequest();

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    v_CallId := v_RegisterReq.msgHeader.callId.callid;

    v_SecurityClientParamsList[0] := f_IMS_Register_GetSecurityClientParams(v_RegisterReq);  /* @sic R5-144754 sic@ */

    // Step 2

    // @siclog "Step 2" siclog@

    /* f_IMS_Register_SecurityInit(v_RegisterReq);       @sic R5s130264 Change 2.1: security needs to be initialised just before (first) 401 response is sent sic@  */

    v_AuthenticationParams := f_IMS_AuthenticationInit(tsc_IMS_AuthRAND);

    f_IMS_Register_SecurityInitPortsAndSPIs(v_RegisterReq, v_AuthenticationParams);   /* @sic R5s131004 - MCC160 implementation  sic@

                                                                                         security needs to be initialised at the IMS PTC otherwise f_IMS_PTC_BuildSecurityServerParam will cause a runtime errror */
    f_IMS_PortsAndSecurityConfig(f_SIP_ContactUrl_GetPort(f_MessageHeader_GetContactSipUrl(
                                                                                                          v_RegisterReq.msgHeader)),                 // may be omit in which case 5060 is chosen by the IP PTC

                                   cs_IMS_RegistrationInfo(f_IMS_PTC_NW_Address_Get(), f_IMS_PTC_UE_Address_Get()));
    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterReq, v_InvalidNonce))));

    // Step 3

    // @siclog "Step 3" siclog@

    v_IMS_DATA_REQ := f_IMS_REGISTER_RequestWithSpecificAuth(v_CallId, cr_Authorization_9_2(v_InvalidNonce, tsc_IMS_Opaque));

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    v_SecurityClientParamsList := fl_CheckSecurityClientParams(v_SecurityClientParamsList, v_RegisterReq);  /* @sic R5-144754 sic@ */

    // Step 4

    // @siclog "Step 4" siclog@

    f_IMS_Register_SecurityInit(v_RegisterReq);       /* @sic R5s130573: IPSec layer needs to be updated acc. to new Security client header sent by the UE sic@ */

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterReq))));

    // Step 5

    // @siclog "Step 5" siclog@

    v_IMS_DATA_REQ := f_IMS_REGISTER_SubsequentRequest();

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);   /* @sic R5s130264 Change 2.2: security has changed to "protected" now sic@ */

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    // Step 6

    // @siclog "Step 6" siclog@

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_RegisterReq))));

    f_IMS_Register_SubscribeNotify(v_RegisterReq);

    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_InitialRegistration);

  }

}


