
	Reason for change:
	The following issues have been found in the altstep a_IMS_REFER_ReceiveRequest(), used to implement step 1 of generic procedure C.19 from 34.229-1 (“UE sends REFER to SS referring to the conference”)

1. The function f_SIP_AmpersandParamList_GetParam() is used to search for the “Replaces” header. This function eventually calls fl_SIP_GenericParamList_GetParamIndex() which searches the generic parameter list by converting the “id” field to lower case before comparing it with the input string. The function is currently called with input “Replaces” which will not find the correct parameter.

2. The current implementation of the altstep compares the dialog id in the “replaces” header with the information for the current active dialog. This is incorrect as the current active dialog will be the conference call. The information should be compared against either of the two dialogs established earlier in the test case which the UE is attempting to add to the conference call.



	s
	

	Summary of change:
	1. Pass the string “replaces” into function f_SIP_AmpersandParamList_GetParam() (instead of “Replaces”).
2. Compare the dialog information in the “replaces” header with the information for both of the two previously established dialogs (firstDialog or secondDialog).
Note1 : Also the designation of the dialog tags has been swapped compared to the original implementation – “from-tag” is the remote tag and “to-tag” is the local tag.

Note2 : Since this altstep can run in parallel with the UE releasing the two earlier dialogs, it is possible the UE can send REFER to replace one dialog, then send BYE to release it, then send REFER to replace the other dialog. So it is necessary to check if each dialog still exists before creating the templates, otherwise the expression e.g. valueof(f_IMS_PTC_ImsInfo_DialogGetRemoteTag() can give a run-time error).

Change 1 
	Function name
	Altstep a_IMS_REFER_ReceiveRequest()

	Reason for change
	The following issues have been found in the altstep a_IMS_REFER_ReceiveRequest(), used to implement step 1 of generic procedure C.19 from 34.229-1 (“UE sends REFER to SS referring to the conference”)

1. The function f_SIP_AmpersandParamList_GetParam() is used to search for the “Replaces” header. This function eventually calls fl_SIP_GenericParamList_GetParamIndex() which searches the generic parameter list by converting the “id” field to lower case before comparing it with the input string. The function is currently called with input “Replaces” which will not find the correct parameter.

2. The current implementation of the altstep compares the dialog id in the “replaces” header with the information for the current active dialog. This is incorrect as the current active dialog will be the conference call. The information should be compared against either of the two dialogs established earlier in the test case which the UE is attempting to add to the conference call.



	Summary of change
	1. Pass the string “replaces” into function f_SIP_AmpersandParamList_GetParam() (instead of “Replaces”).
2. Compare the dialog information in the “replaces” header with the information for wither of the two previously established dialogs (firstDialog or secondDialog).
Note1 : Also the designation of the dialog tags has been swapped compared to the original implementation – “from-tag” is the remote tag and “to-tag” is the local tag.

Note2 : Since this altstep can run in parallel with the UE releasing the two earlier dialogs, it is possible the UE can send REFER to replace one dialog, then send BYE to release it, then send REFER to replace the other dialog. So it is necessary to check if each dialog still exists before creating the templates, otherwise the expression e.g. valueof(f_IMS_PTC_ImsInfo_DialogGetRemoteTag() can give a run-time error).

	TTCN module
	IMS_CC_SSConferencingtestcases.ttcn

	MCC160 Comment
	


Before change

    altstep a_IMS_REFER_ReceiveRequest(out IMS_DATA_REQ p_IMS_REFER_REQ,

                                     charstring p_RequestUriString,

                                     charstring p_ReferToUriString,

                                     boolean p_ActiveSessionToBeReplaced,

                                     template (present) integer p_CSeqValue := ?,

                                     boolean p_ReferCreatingDialog := false) runs on IMS_PTC

  { /* @sic R5s141343 change 3.1: p_ReferCreatingDialog := false sic@ */

    /* @sic R5-153765: new parameter p_ActiveSessionToBeReplaced sic@

     * NOTE: p_ActiveSessionToBeReplaced==true => there is an active session between the UE and a remote user which shall be replaced (e.g. Three way session creation; 15.121a);

     *       p_ActiveSessionToBeReplaced==false => there is no active session with the remote user but the REFER is just ot invite the remote user */

    var IMS_DATA_REQ v_IMS_REFER_REQ;

    var template (present) SipUrl v_RequestUriRefer := f_SIP_BuildSipUri_RX(p_RequestUriString);

    var template AmpersandParam_List v_Headers := *;   /* @sic R5-151798: SIP URI may contain a "Replaces" header referring to the dialog ID which has been established before sic@ */

    var template (present) SipUrl v_ReferToUri := f_SIP_BuildSipUri_RX(p_ReferToUriString, -, -, v_Headers);  /* @sic R5s150382 change 3 sic@ */

    var template (present) MessageHeader v_MessageHeader_Refer := f_IMS_ReferRequest_MessageHeaderRX(v_RequestUriRefer, v_ReferToUri, p_ReferCreatingDialog, p_CSeqValue);

    var template (omit) GenericParam v_Replaces;

    var template (value) CallId v_CallId;
    var CharStringList_Type v_ReceivedDialogId;

    var template (present) CharStringList_Type v_ExpectedDialogId;
    var integer v_CSeqValue;

    [] IMS_Server.receive(car_IMS_Refer_Request(cr_REFER_Request(v_RequestUriRefer, v_MessageHeader_Refer))) -> value v_IMS_REFER_REQ

      {

        f_IMS_A_2_10_Refer_CommonChecks(v_IMS_REFER_REQ, p_ReferCreatingDialog);

        if (not isvalue(p_CSeqValue)) {       /* when there are multiple REFERs in parallel, we cannot determine CSeq in advance */

          v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogRemote);                      /* => get next CSeq value; CSeq gets incremented for the remote (UE) side of the current dialog */

          f_IMS_MessageHeader_CheckCSeq(v_IMS_REFER_REQ.Request.Refer.msgHeader, v_CSeqValue);   /* @sic R5-151954 sic@ */

        }

        // check whether header portion of the SIP URI includes Replaces header if an active session exists @sic R5-153765 sic@

        v_Replaces := f_SIP_AmpersandParamList_GetParam(v_IMS_REFER_REQ.Request.Refer.msgHeader.referTo.nameAddr.addrSpec.headers, "Replaces");

        if (p_ActiveSessionToBeReplaced) {   /* @sic R5-153765 sic@ */

          if (ispresent(v_Replaces)) {

            v_CallId := f_IMS_PTC_ImsInfo_DialogGetCallId();

            v_ExpectedDialogId := {

              v_CallId.callid,

              permutation("to-tag=" & valueof(f_IMS_PTC_ImsInfo_DialogGetRemoteTag()), "from-tag=" & valueof(f_IMS_PTC_ImsInfo_DialogGetLocalTag()), *)

            };
            v_ReceivedDialogId := f_StringSplit(f_URL_Decoding(valueof(v_Replaces.paramValue.tokenOrHost)), {";"}); // see RFC 3891 6.1

            if (not match(v_ReceivedDialogId, v_ExpectedDialogId)) {
              f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid dialog id in Replaces header of Refer-To's SipUri");

            }

          } else {

            f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Refer-To does not include Replaces header even though there an active session");

          }

        } else {

          if (ispresent(v_Replaces)) {

            f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Refer-To includes Replaces header even though there no active session");

          }

        }

        p_IMS_REFER_REQ := v_IMS_REFER_REQ;  // return message by ref

      }

  }

After change

  altstep a_IMS_REFER_ReceiveRequest(out IMS_DATA_REQ p_IMS_REFER_REQ,

                                     charstring p_RequestUriString,

                                     charstring p_ReferToUriString,

                                     boolean p_ActiveSessionToBeReplaced,

                                     template (present) integer p_CSeqValue := ?,

                                     boolean p_ReferCreatingDialog := false) runs on IMS_PTC

  { /* @sic R5s141343 change 3.1: p_ReferCreatingDialog := false sic@ */

    /* @sic R5-153765: new parameter p_ActiveSessionToBeReplaced sic@

     * NOTE: p_ActiveSessionToBeReplaced==true => there is an active session between the UE and a remote user which shall be replaced (e.g. Three way session creation; 15.121a);

     *       p_ActiveSessionToBeReplaced==false => there is no active session with the remote user but the REFER is just ot invite the remote user */

    var IMS_DATA_REQ v_IMS_REFER_REQ;

    var template (present) SipUrl v_RequestUriRefer := f_SIP_BuildSipUri_RX(p_RequestUriString);

    var template AmpersandParam_List v_Headers := *;   /* @sic R5-151798: SIP URI may contain a "Replaces" header referring to the dialog ID which has been established before sic@ */

    var template (present) SipUrl v_ReferToUri := f_SIP_BuildSipUri_RX(p_ReferToUriString, -, -, v_Headers);  /* @sic R5s150382 change 3 sic@ */

    var template (present) MessageHeader v_MessageHeader_Refer := f_IMS_ReferRequest_MessageHeaderRX(v_RequestUriRefer, v_ReferToUri, p_ReferCreatingDialog, p_CSeqValue);

    var template (omit) GenericParam v_Replaces;

    var template (value) CallId v_CallId1;

    var template (value) CallId v_CallId2;
    var CharStringList_Type v_ReceivedDialogId;

    var template (present) CharStringList_Type v_ExpectedDialogId1:=omit;

    var template (present) CharStringList_Type v_ExpectedDialogId2:=omit;
    var integer v_CSeqValue;

    [] IMS_Server.receive(car_IMS_Refer_Request(cr_REFER_Request(v_RequestUriRefer, v_MessageHeader_Refer))) -> value v_IMS_REFER_REQ

      {

        f_IMS_A_2_10_Refer_CommonChecks(v_IMS_REFER_REQ, p_ReferCreatingDialog);

        if (not isvalue(p_CSeqValue)) {       /* when there are multiple REFERs in parallel, we cannot determine CSeq in advance */

          v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogRemote);                      /* => get next CSeq value; CSeq gets incremented for the remote (UE) side of the current dialog */

          f_IMS_MessageHeader_CheckCSeq(v_IMS_REFER_REQ.Request.Refer.msgHeader, v_CSeqValue);   /* @sic R5-151954 sic@ */

        }

        // check whether header portion of the SIP URI includes Replaces header if an active session exists @sic R5-153765 sic@

        v_Replaces := f_SIP_AmpersandParamList_GetParam(v_IMS_REFER_REQ.Request.Refer.msgHeader.referTo.nameAddr.addrSpec.headers, "replaces");

        if (p_ActiveSessionToBeReplaced) {   /* @sic R5-153765 sic@ */

          if (ispresent(v_Replaces)) {

            v_CallId1 := f_IMS_PTC_ImsInfo_DialogGetCallId(firstDialog);

            v_CallId2 := f_IMS_PTC_ImsInfo_DialogGetCallId(secondDialog);

            if (not (f_IMS_PTC_ImsInfo_DialogIsReleased(firstDialog)))

            {

               v_ExpectedDialogId1 := {

                v_CallId1.callid,

                permutation("to-tag=" & valueof(f_IMS_PTC_ImsInfo_DialogGetLocalTag(firstDialog)), "from-tag=" & valueof(f_IMS_PTC_ImsInfo_DialogGetRemoteTag(firstDialog)), *)

              };

            }

            if (not (f_IMS_PTC_ImsInfo_DialogIsReleased(secondDialog)))

            {

              v_ExpectedDialogId2 := {

                v_CallId2.callid,

                permutation("to-tag=" & valueof(f_IMS_PTC_ImsInfo_DialogGetLocalTag(secondDialog)), "from-tag=" & valueof(f_IMS_PTC_ImsInfo_DialogGetRemoteTag(secondDialog)), *)

              };

            }
            v_ReceivedDialogId := f_StringSplit(f_URL_Decoding(valueof(v_Replaces.paramValue.tokenOrHost)), {";"}); // see RFC 3891 6.1

            if (not match (v_ReceivedDialogId, v_ExpectedDialogId1) and not match (v_ReceivedDialogId, v_ExpectedDialogId2))  {
              f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid dialog id in Replaces header of Refer-To's SipUri");

            }

          } else {

            f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Refer-To does not include Replaces header even though there an active session");

          }

        } else {

          if (ispresent(v_Replaces)) {

            f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Refer-To includes Replaces header even though there no active session");

          }

        }

        p_IMS_REFER_REQ := v_IMS_REFER_REQ;  // return message by ref

      }

  }
